# $Id$
# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>

pkgname=dleyna-core
pkgver=0.6.0
pkgrel=1
pkgdesc="Library of utility functions that are used by the higher level dLeyna"
arch=(i686 x86_64)
url="http://01.org/dleyna"
license=(LGPL2.1)
depends=(gupnp)
makedepends=(git)
_commit=32207688478b890c91eb084a84f96b40a729ab14  # tags/v0.6.0
source=("git+https://github.com/01org/dleyna-core#commit=$_commit"
        crash.patch)
sha256sums=('SKIP'
            '28f8fa43476e4c71b8b1f16b7ab47df045db3c8490a97c19c7354ce56094bffe')

pkgver() {
  cd $pkgname
  git describe --tags | sed 's/^v//;s/-/+/g'
}

prepare() {
  cd $pkgname
  NOCONFIGURE=1 ./autogen.sh
}

build() {
  cd $pkgname
  ./configure --prefix=/usr --sysconfdir=/etc --localstatedir=/var \
    --disable-static
  make
}

check() {
  cd $pkgname
  make check
}

package() {
  cd $pkgname
  make DESTDIR="$pkgdir" install
}
